


## 18.MİKROİŞLEMCİ TEMELLİ SİSTEM UYGULAMALARI

Mikroişlemci temelli sistemler eğitim, güvenlik, ticari, endüstriyel, askeri, sağlık, vs. alanlardaki uygulamalarda yaygın olarak kullanılmaktadır. Bunun sonucunda uygulamaya yönelik olarak mikroişlemcili sistemlerin donanım ve yazılımının tasarlanması ve gerçekleştirilmesi çeşitli alanlardaki mühendislik çalışmalarında önemli bir yer tutmaktadır. Endüstriyel uygulamalar, kendisi de mikroişlemci temelli bir sistem olan kişisel bilgisayarlar ile gerçekleştirilebilmektedir.

### 18.1. Giriş Uygulamaları

Mikroişlemci temelli bir sisteme dışarıdan bilgi girişi yapmanın en basit yöntemi yapılan işe uygun bir anahtar kullanmaktır. Şekil 18-1'de [tek kutup çift yönlü \(Single Pole Double Trough, SPDT\)](#) bir anahtar kullanılarak tasarlanmış giriş devresi ve blok diyagramı verilmiştir. Burada anahtar önce **normalde kapalı (Normally Close, NC)** kontaklıdır. Bu durumda +5V'a bağlı olan direnç üzerinden çınlama önleyici devreye lojik "1" seviyesinde gerilim giriş olarak uygulanır. Sonra bu anahtar kapatılmasıyla **normalde açık (Normally Open, NO)** kontak durumuna geçer.



Şekil 18-1 Bir anahtar ile oluşturulmuş fiziksel giriş devresi

**Örnek 18-1** Şekil 18-1'de verilen şekilde 6821 PIA tümleşik devresinden oluşan sayısal giriş biriminde PA7 ucuna bağlı bir anahtarın açık/kapalı olma durumuna göre çalışan 6802 mikroişlemci dilinde bir programı, çevirici çıkış dosyası şeklinde yazınız.


#### Çözüm:

```

F00F B68000 ANA: LDAA PIADRA ;Sayısal giriş biriminin okunması
F012 8480 ANDA #10000000B ;Giriş birimi ucunun tümleyeni Z bayrağına alınır.
F014 2706 BEQ S7K ;PA7 ucu "0" ise anahtar kapalıdır ve Z=1 olur.
F016 ... ;Anahtar "7" açıksa yapılacak işlemler burada yer alır.
F019 7EF00F JMP ANA ;Ana program döngüsüne git
F01C S7K: ... ;Anahtar "7" kapalıysa yapılacak işlemler burada yer alır.
F01F 7EF00F JMP ANA ;Ana program döngüsüne git

```





Şekil 18-4 Çok konumlu anahtar ile oluşturulmuş fiziksel giriş devresi

**Örnek 18-3** Şekil 18-4'de verilen şekilde 6821 PIA tümleşik devresinden oluşan sayısal giriş biriminde A portuna doğrudan bağlı olan 8 konumlu (1 kutuplu 8 yollu) komütatör anahtarın konumunu elde etmek için 6802 mikroişlemci dilinde bir programı, çevirici çıkış dosyası şeklinde yazınız.

**Çözüm:**

**Programın çıkış dosyası:**

```

8000 = PIADRA: EQU 8000H ;PIA Port A Veri/Veri Yönü Yazmaç adresi
8001 = PIACRA: EQU 8001H ;PIA Port A Kontrol Yazmaç (CRA) adresi
F000 ORG 0F000H
F000 8E007F BASLA: LDS #7FH ;Yığın bölgesini iç RAM'in tepe adresine ayarla
F003 7F8001 CLR PIACRA ;PIA Port A Veri Yönü Yazmacının seçilmesi (CRA b2="0")
F006 4F CLRA
F007 B78000 STAA PIADRA ;PIA Port A 8-Bit Giriş (PA0-PA7=Giriş) DDRA=00H
F00A 8604 LDAA #04H
F00C B78001 STAA PIACRA ;PIA Port A Veri Yazmacının seçilmesi (CRA b2="1")
F00F ANA: ;ana program: ilk koşullar...
; ... ;ana program: genel işlemler...
F023 BDF100 JSR BEK1 ;çınlamanın geçmesi için bekleme altprogramı
F026 C608 LDAB #8 ;Anahtar konumu sayacı (burada 8 anahtar var)
F028 B68000 LDAA PIADRA ;Sayısal giriş biriminin okunması
F02B 48 L1: ASLA ;Giriş biriminin ucunun durumu C bayrağına alınır.
F02C 2409 BCC KAP7 ;uç "0" ise anahtar kapatılmıştır.
F02E 5A DECB ;anahtar konumu sayacını azalt
F02F 26FA BNE L1 ; anahtarın 8 konumu bitene kadar devam.
F031 BDF200 JSR HATA ;anahtar açık ise yapılacaklar burada yer alır.
F034 7EF003 JMP ANA ;ana program döngüsüne git
F037 C108 KAP7: CMPB #8 ;anahtarın konumu "7" mi?
F039 2606 BNE KAP6
F03B BDE700 JSR PRG7 ; "7" ise yapılacak işlemler.
F03E 7EF003 JMP ANA ;ana program döngüsüne git
F041 C107 KAP6: CMPB #7 ;anahtarın konumu "6" mı?
F043 2606 BNE KAP5
F045 BDE600 JSR PRG6 ; "6" ise yapılacak işlemler.
F048 7EF003 JMP ANA ;ana program döngüsüne git
F04B C106 KAP5: CMPB #6 ;anahtarın konumu "5" mi?
F04D 2606 BNE KAP4

```

```

F04F BDE500      JSR   PRG5   ; "5" ise yapılacak işlemler.
F052 7EF003      JMP   ANA    ; ana program döngüsüne git
F055 C105      KAP4:  CMPB  #5    ; anahtarın konumu "4" mı?
F057 2606      BNE   KAP3
F059 BDE400      JSR   PRG4   ; "4" ise yapılacak işlemler.
F05C 7EF003      JMP   ANA    ; ana program döngüsüne git
F05F C104      KAP3:  CMPB  #4    ; anahtarın konumu "3" mı?
F061 2606      BNE   KAP2
F063 BDE300      JSR   PRG3   ; "3" ise yapılacak işlemler.
F066 7EF003      JMP   ANA    ; ana program döngüsüne git
F069 C103      KAP2:  CMPB  #3    ; anahtarın konumu "2" mı?
F06B 2606      BNE   KAP1
F06D BDE200      JSR   PRG2   ; "2" ise yapılacak işlemler.
F070 7EF003      JMP   ANA    ; ana program döngüsüne git
F073 C102      KAP1:  CMPB  #2    ; anahtarın konumu "1" mi?
F075 2606      BNE   KAP0
F077 BDE100      JSR   PRG1   ; "1" ise yapılacak işlemler.
F07A 7EF003      JMP   ANA    ; ana program döngüsüne git
F07D BDE000      KAP0:  JSR   PRG0   ; "0" ise yapılacak işlemler.
F080 7EF003      JMP   ANA    ; ana program döngüsüne git
; 10ms gecikme sağlayan altprogram, sistem saati 1 MHz ise 10000 x 1us =10ms.
; Toplam gecikme = 3 + 1249 x (4+4) + 5 = 10000 sistem saati
F100              ORG   0F100H
F100 CE04E1      BEK1:  LDX   #1249 ;döngü sayacının ilk değeri
F103 09          L2:    DEX           ;döngü sayacının
F104 26FD          BNE   L2           ;sıfır olana kadar azaltılması.
F106 39          RTS            ;ana programa geri dönülmesi.
;Komütatör anahtarın arızalı olması durumunda yapılacak işlemler altprogramı.
F200              ORG   0F200H
F200 39          HATA:  RTS
;Altprogramlar
E000              ORG   0E000H
E000 39          PRG0:  RTS
E100              ORG   0E100H
E100 39          PRG1:  RTS
E200              ORG   0E200H
E200 39          PRG2:  RTS
E300              ORG   0E300H
E300 39          PRG3:  RTS
E400              ORG   0E400H
E400 39          PRG4:  RTS
E500              ORG   0E500H
E500 39          PRG5:  RTS
E600              ORG   0E600H
E600 39          PRG6:  RTS
E700              ORG   0E700H
E700 39          PRG7:  RTS
;Vektör Adresleri
FFFE              ORG   0FFFEH
FFFE F000        DWM   BASLA ;RES Reset, Mikroişlemciyi Yeniden Başlatma Adresi
0000              END

```

**Örnek 18-4** Şekil 18-4'de verilen şekilde 6821 PIA tümleşik devresinden oluşan sayısal giriş biriminde A portuna doğrudan bağlı olan 8 konumlu (1 kutuplu 8 yollu) komütatör anahtarın konumu ASCII olarak elde edilecektir. Tabloya bakma (lookup table) yöntemiyle çalışan 6802 mikroişlemci dilinde bir programı, çevirici çıkış dosyası şeklinde yazınız.

### Çözüm:

#### Programın çıkış dosyası:


```

8000 =      PIADRA: EQU   8000H ;PIA Port A Veri/Veri Yönü Yazmaç adresi
8001 =      PIACRA: EQU   8001H ;PIA Port A Kontrol Yazmaç (CRA) adresi
F000                ORG   0F000H
F000 8E007F  BASLA: LDS   #7FH   ;Yığın bölgesini iç RAM'in tepe adresine ayarla
F003 7F8001                CLR   PIACRA ;PIA Port A Veri Yönü Yazmacının seçilmesi (CRA b2="0")
F006 4F                CLRA
F007 B78000                STAA  PIADRA ;PIA Port A 8-Bit Giriş (PA0-PA7=Giriş) DDRA=00H
F00A 8604                LDAA  #04H
F00C B78001                STAA  PIACRA ;PIA Port A Veri Yazmacının seçilmesi (CRA b2="1")
F00F                ANA:
;                ;ana program: program için ilk koşullar...
;                ;ana program: genel işlemler...
F023 BDF100                JSR   BEK1 ;çınlamanın geçmesi için bekleme altprogramı
F026 CE0008                LDX  #8   ;Anahtar konumu sayacı (burada 8 anahtar var)
F029 B68000                LDAA  PIADRA ;Sayısal giriş biriminin okunması
F02C 48                L1:  ASLA ;Giriş biriminin ucunun durumu C bayrağına alınır.
F02D 2409                BCC  KAPX ;uç "0" ise anahtar kapatılmıştır.
F02F 09                DEX   ;anahtar konumu sayacını azalt
F030 26FA                BNE  L1 ; anahtarın 8 konumu bitene kadar devam.
F032 BDF200                JSR  HATA ;anahtar açık ise yapılacaklar burada yer alır.
F035 7EF00F                JMP  ANA ;ana program döngüsüne geri dön
F038 09                KAPX: DEX
F039 A660                LDAA  TUS,X ;anahtarın ASCII kodu ile
F03B 7EF00F                JMP  ANA ;ana program döngüsüne geri dön
;TUS adlarının bulunduğu tablo
0060                ORG   60H
0060 30313233  TUS:  DFB  30H,31H,32H,33H
0064 34353637                DFB  34H,35H,36H,37H
;10 ms gecikme sağlayan altprogram, sistem saati 1 MHz ise 10000 x 1us = 10 ms.
;6802 için toplam gecikme = 3 + 1249 x (4+4) + 5 = 10000 sistem saati
F100                ORG   0F100H
F100 CE04E1  BEK1: LDX  #1249 ;döngü sayacının ilk değeri
F103 09                L2:  DEX   ;döngü sayacının
F104 26FD                BNE  L2 ;sıfır olana kadar azaltılması.
F106 39                RTS   ;ana programa geri dönülmesi.
;Komütatör anahtarın arızalı olması durumunda yapılacak işlemler altprogramı.
F200                ORG   0F200H
F200 39                HATA: RTS
;Vektör Adresleri
FFFE                ORG   0FFFEH
FFFE F000                DWM  BASLA ;RES Reset, Mikroişlemciyi Yeniden Başlatma Adresi
0000                END

```

## 18.2. Çıkış Uygulamaları


Mikroişlemci temelli bir sistemden dışarıya bilgi çıkışı yapmanın en basit yöntemi yapılan işe uygun bir gösterge kullanmaktır. Bu göstergeler gerçekleştirilen ölçüm, denetlenen süreç hakkında sayısal değerleri gösteren genellikle ekonomik ve uzun ömürlü olması nedeniyle ışık yayan diyot (LED) veya sıvı kristal yapıda olabilir.



Şekil 18-5 LED kullanarak yapılan çıkış uygulamaları

Sürücü çıkışındaki akımı sınırlamak için LED'e seri R direnci kullanılır. Bu direncin hesabı için Şekil 18-6 ve Şekil 18-7'de verilen eşdeğer devreler kullanılabilir.

$$R1 = \frac{V_{OH} - V_{LEDON}}{I_{LEDON}} = \frac{3.6 - 1.6}{20 \cdot 10^{-3}} = 100\Omega$$




Şekil 18-6 LED'in pozitif lojik ile çalıştırılması için eşdeğer devre

Tablo 18-1 Değişik renkteki LED'lerin DC özellikleri


LED	VLEDON	ILEDON
Düşük Akımlı Kırmızı	1.8 V	2 mA
Standart Kırmızı	1.6 V	20 mA
Parlak Kırmızı	2.2 V	20 mA
Standart Sarı	2.2 V	10 mA
Standart Yeşil	2.3 V	10 mA

Şekil 18-7'de verilen devrede LED'in negatif lojik ile sürülmesi durumunda bir eşdeğer devre verilmiştir.




Şekil 18-7 LED'in negatif lojik ile çalıştırılması için eşdeğer devre


$$R1 = \frac{V_{CC} - V_{OL} - V_{LEDON}}{I_{LEDON}} = \frac{5 - 0.4 - 1.6}{20 \cdot 10^{-3}} = 150\Omega$$



Şekil 18-8 7-parça LED gösterge parça tanımları



Şekil 18-9 7-parça LED gösterge için Ortak Katot ve Ortak Anot bağlantı şekli




Şekil 18-10 Paralel G/Ç tümleşik devresi ile 7-parça LED gösterge parça bağlantısı

Tablo 18-2 7-parça gösterge için veri dönüşüm tablosu

Sayı	Ortak Anot			Ortak Katot			
	g	f	e d c b a	hex	g	f e d c b a	hex
0	1	0	0 0 0 0 0	40	0	1 1 1 1 1 1	3F
1	1	1	1 0 0 1	79	0	0 0 0 1 1 0	06
2	0	1	0 0 1 0 0	29	1	0 1 1 0 1 1	5B
3	0	1	1 0 0 0 0	30	1	0 0 1 1 1 1	4F
4	0	0	1 1 0 0 1	19	1	1 1 0 0 1 1 0	66
5	0	0	1 0 0 1 1	13	1	1 1 0 1 1 0 0	6C
6	0	0	0 0 0 1 1	03	1	1 1 1 1 1 0 0	7C
7	1	1	1 1 0 0 0	78	0	0 0 0 1 1 1	07
8	0	0	0 0 0 0 0	00	1	1 1 1 1 1 1 1	7F
9	0	0	1 1 0 0 0	18	1	1 1 0 0 1 1 1	67





Şekil 18-11 7-parça LED göstergede sayıların şekil ve tanımları

**Örnek 18-5** Şekil 18-10'deki blok diyagramda olduğu gibi 6821 PIA kullanarak B çıkış portuna bağlanan ortak katot 7-parça LED göstergeye "0" ve "1" yazılması için iki ayrı programı 6802 mikroişlemci dilinde, bir çevirici çıkış dosyası şeklinde yazınız.

**Çözüm:**

**Programların çıkış dosyaları:**

8002 =	PIADR: EQU	8002H	;PIA Port B Veri/Veri Yönü (DRB/DDRB) Yazmaç adresi
8003 =	PIACRB: EQU	8003H	;PIA Port B Kontrol Yazmaç (CRB) adresi
F000	ORG	0F000H	
F000 8E007F	BASLA: LDS	#7FH	;Yığın bölgesini iç RAM'in tepe adresine ayarla
F003 7F8003	CLR	PIACRB	;PIA Port B Veri Yönü Yazmacının seçilmesi (CRB b2="0")
F006 86FF	LDAA	#0FFH	
F008 B78002	STAA	PIADR: B	;PIA Port B 8-Bit Çıkış (PB0-PB7=Çıkış) DDRB=00H
F00B 8604	LDAA	#04H	
F00D B78003	STAA	PIACRB: B	;PIA Port B Veri Yazmacının seçilmesi (CRB b2="1")
F010 863F	ANA: LDAA	#3FH	;göstergeye "0" bilgisi yazılması için kod
F012 B78002	STAA	PIADR: B	;Sayısal çıkış biriminin B portuna yazılması
F010 8606	ANA: LDAA	#06H	;göstergeye "1" bilgisi yazılması için kod
F012 B78002	STAA	PIADR: B	;Sayısal çıkış biriminin B portuna yazılması
F015 7EF010	JMP	ANA	;Ana program döngüsüne git
0000	END		

**Örnek 18-6** Şekil 18-10'deki blok diyagramı kullanarak Paralel Giriş/Çıkış Birimine (6821,PIA) bağlanan ortak katot 7-parça LED göstergede, 1 saniye süreli aralıkla sırayla 0'dan 9'a kadar olan sayıların gösterilmesi işlemi bir BCD sayıcı gibi sürekli olarak yapılacak şekilde gösterilecektir. Programı 6802 mikroişlemcisinin dilinde, bir çevirici çıkış dosyası biçiminde olacak şekilde gerekli olan bütün tanımlamaları ve her satırındaki komutun açıklamasını yaparak yazınız.

**Çözüm:****Programın çıkış dosyası:**

```
8002 =      PIADB: EQU   8002H ;PIA Port B Veri/Veri Yönü (DRB/DDR) Yazmaç adresi
8003 =      PIACRB: EQU  8003H ;PIA Port B Kontrol Yazmaç (CRB) adresiF000
           ORG   0F000H
F000 8E007F BASLA: LDS   #7FH ;Yığın bölgesini iç RAM'in tepe adresine ayarla
F003 7F8003 CLR   PIACRB ;PIA Port B Veri Yönü Yazmacının seçilmesi (CRB b2="0")
F006 86FF   LDAA  #0FFH
F008 B78002 STAA  PIADB ;PIA Port B 8-Bit Çıkış (PB0-PB7=Çıkış) DDRB=00H
F00B 8604   LDAA  #04H
F00D B78003 STAA  PIACRB ;PIA Port B Veri Yazmacının seçilmesi (CRB b2="1")
```

F010 863F	ANA:	LDAA	#3FH	;göstergeye "0" bilgisi için kod
F012 B78002		STAA	PIADB	;Sayısal çıkış biriminin B portuna yazılması
F015 BDF100		JSR	BEK1	;yazılanı gözün algılaması için bekleme altprogramı
F018 8606		LDAA	#06H	;göstergeye "1" bilgisi için kod
F01A B78002		STAA	PIADB	;Sayısal çıkış biriminin B portuna yazılması
F01D BDF100		JSR	BEK1	;yazılanı gözün algılaması için bekleme altprogramı
F020 865B		LDAA	#5BH	;göstergeye "2" bilgisi için kod
F022 B78002		STAA	PIADB	;Sayısal çıkış biriminin B portuna yazılması
F025 BDF100		JSR	BEK1	;yazılanı gözün algılaması için bekleme altprogramı
F028 864F		LDAA	#4FH	;göstergeye "3" bilgisi için kod
F02A B78002		STAA	PIADB	;Sayısal çıkış biriminin B portuna yazılması
F02D BDF100		JSR	BEK1	;yazılanı gözün algılaması için bekleme altprogramı
F030 8666		LDAA	#66H	;göstergeye "4" bilgisi için kod
F032 B78002		STAA	PIADB	;Sayısal çıkış biriminin B portuna yazılması
F035 BDF100		JSR	BEK1	;yazılanı gözün algılaması için bekleme altprogramı
F038 866C		LDAA	#6CH	;göstergeye "5" bilgisi için kod
F03A B78002		STAA	PIADB	;Sayısal çıkış biriminin B portuna yazılması
F03D BDF100		JSR	BEK1	;yazılanı gözün algılaması için bekleme altprogramı
F040 867C		LDAA	#7CH	;göstergeye "6" bilgisi için kod
F042 B78002		STAA	PIADB	;Sayısal çıkış biriminin B portuna yazılması
F045 BDF100		JSR	BEK1	;yazılanı gözün algılaması için bekleme altprogramı
F048 8607		LDAA	#07H	;göstergeye "7" bilgisi için kod
F04A B78002		STAA	PIADB	;Sayısal çıkış biriminin B portuna yazılması
F04D BDF100		JSR	BEK1	;yazılanı gözün algılaması için bekleme altprogramı
F050 867F		LDAA	#7FH	;göstergeye "8" bilgisi için kod
F052 B78002		STAA	PIADB	;Sayısal çıkış biriminin B portuna yazılması
F055 BDF100		JSR	BEK1	;yazılanı gözün algılaması için bekleme altprogramı
F058 8667		LDAA	#67H	;göstergeye "9" bilgisi için kod
F05A B78002		STAA	PIADB	;Sayısal çıkış biriminin B portuna yazılması
F05D BDF100		JSR	BEK1	;yazılanı gözün algılaması için bekleme altprogramı
F060 7EF010		JMP	ANA	;Ana program döngüsüne git

```
        ;1 sn gecikme sağlayan altprogram, sistem saati 32.768 kHz ise
        ;6802 için toplam gecikme = 3 + 4095 x (4+4) + 5 = 32768 sistem saati
F100          ORG  0F100H
F100 CE0FFF  BEK1:  LDX  #4095  ;döngü sayacının ilk değeri
F103 09      L2:    DEX          ;döngü sayacının
F104 26FD          BNE  L2      ;sıfır olana kadar azaltılması.
F106 39          RTS          ;ana programa geri dönülmesi.

        ;Vektör Adresleri
FFFE          ORG  0FFFEH
FFFE F000     DWM  BASLA  ;RES Reset, Mikroişlemciyi Yeniden Başlatma Adresi
0000          END
```

**Örnek 18-7** Şekil 18-10'deki blok diyagramı kullanarak ortak katot 7-parça LED göstergeye çıkış işlemini gerçekleştirmek üzere tasarlanmış bir programı tabloya bakma yöntemiyle çalışan 6802 mikroişlemci dilinde, çevirici çıkış dosyası şeklinde yazınız.

**Çözüm:****Programın çıkış dosyası:**

```

8002 =      PIADR: EQU   8002H   ;PIA Port B Veri/Veri Yönü (DRB/DDR) Yazmaç adresi
8003 =      PIACR: EQU   8003H   ;PIA Port B Kontrol Yazmaç (CRB) adresi
0000      ORG   0000H
0000      TMP:   DFS   2       ;Geçici veri için bellekte 2 Bayt yer ayır
F000      ORG   0F000H
F000 8E007F BASLA: LDS   #7FH   ;Yığın bölgesini iç RAM'in tepe adresine ayarla
F003 7F8003 CLR   PIACR   ;PIA Port B Veri Yönü Yazmacının seçilmesi (CRB b2="0")
F006 86FF   LDAA  #0FFH
F008 B78002 STAA  PIADR   ;PIA Port B 8-Bit Çıkış (PB0-PB7=Çıkış) DDRB=00H
F00B 8604   LDAA  #04H
F00D B78003 STAA  PIACR   ;PIA Port B Veri Yazmacının seçilmesi (CRB b2="1")

```

F010 C60A	ANA:	LDAB	#10	;göstergeye yazılacak sayı adedi
F012 CEF200		LDX	#GTBL	;gösterge değerleri için tablonun başlangıç adresi
F015 A600	L1:	LDAA	0,X	;gösterge bilgisi için kodun tablodan alınması
F017 B78002		STAA	PIADR	;Sayısal çıkış biriminin B portuna yazılması
F01A DF00		STX	TMP	;X dizin yazmacının geçici bellekte saklanması
F01C BDF100		JSR	BEK1	;yazılanı gözün algılaması için bekleme altprogramı
F01F DE00		LDX	TMP	;X dizin yazmacının geçici bellekten geri alınması
F021 08		INX		;bir sonraki gösterge bilgisi adresi
F022 5A		DECB		;sayı adedi sayacı azalt
F023 26F0		BNE	L1	;sayıların gösterilmesi bitene kadar devam.
F025 7EF010		JMP	ANA	;ana program döngüsüne geri dön

;1 sn gecikme sağlayan altprogram, sistem saati 32.768 kHz ise

;6802 için toplam gecikme = 3 + 4095 x (4+4) + 5 = 32768 sistem saati

```

F100      ORG   0F100H
F100 CE0FFF BEK1: LDX   #4095   ;döngü sayacının ilk değeri
F103 09    L2:   DEX           ;döngü sayacının
F104 26FD   BNE   L2         ;sıfır olana kadar azaltılması.
F106 39    RTS             ;ana programa geri dönülmesi.

```

;Ortak katot gösterge kodlarının bulunduğu tablo

```

F200      ORG   0F200H
F200 3F065B4F66 GTBL: DFB  3FH,06H,5BH,4FH,66H ;"0","1","2","3","4"
F205 6C7C077F67 DFB  6CH,7CH,07H,7FH,67H ;"5","6","7","8","9"
F20A 7EF010 JMP   ANA             ;Ana program döngüsüne git

```

;Vektör Adresleri

```

FFFE      ORG   0FFFEH
FFFE F000 DWM  BASLA ;RES Reset, Mikroişlemciyi Yeniden Başlatma Adresi
0000      END

```

## 18.3. Zamanlama Uygulamaları

Mikroişlemci temelli sistemlerin doğadaki olayların değişim hızından çok daha hızlı çalışması nedeniyle, giriş/çıkış biriminin giriş işaretlerini algılaması ve ürettiği çıkış işaretleri çok hızlı değişir. Bu nedenle mikroişlemci yazılımında gecikme programları kullanılarak işaretlerin algılama ve üretim hızları yavaşlatılır.

**Örnek 18-8** Yazılım ile 1,5ms ve 10ms gecikme sağlayan altprogramlar, 4MHz osilatör hızında çalışan 6802 mikroişlemcisinin dilinde tasarlanacaktır. Programı bir çevirici kaynak dosyası biçiminde olacak şekilde gerekli olan bütün tanımlamaları ve her satırındaki komutun açıklamasını yaparak yazınız.

### Çözüm:

4MHz osilatör frekansında çalışan bir 6802 mikroişlemcisi 1MHz sistem saati frekansı ile çalışır. Sistem saati periyodu  $1\mu s$  olduğundan  $1,5ms=1500\mu s$  gecikmeyi sağlamak için 1500 sistem saati süre geçmesi gerekir.

### Programların kaynak dosyaları:

;1.5 ms gecikme sağlayan altprogram.

;f<sub>XTAL</sub>= 4 MHz ise Sistem saati= 1 MHz ve gecikme süresi 1,5 ms = 1500 x 1us

;6802 için toplam gecikme = 3 + TE (toplam sistem saati) x (4+4) + 5 = 1500

;TE = [1500 - (3+5)] / (4+4) = 186,5

;4 MHz f<sub>XTAL</sub> için gerçek gecikme = 3 + 186 x (4+4) + 5 = 1496 sistem saati

; = 1496 x 1 us = 1,496 us  $\cong$  1,5 ms

G1500us: LDX #186 ;döngü sayacının ilk değeri

L2: DEX ;döngü sayacının

BNE L2 ;sıfır olana kadar azaltılması.

RTS ;ana programa geri dönülmesi.

;10 ms gecikme sağlayan altprogram.

;f<sub>XTAL</sub>= 4 MHz ise Sistem saati=1MHz ve gecikme süresi 10 ms = 10000 x 1us

;6802 için toplam gecikme = 3 + TE (toplam sistem saati) x (4+4) + 5 = 10000

;TE = [10000 - (3+5)] / (4+4) = 1249

;4 MHz f<sub>XTAL</sub> için gerçek gecikme = 3 + 1249 x (4+4) + 5 = 10000 sistem saati


; = 10000 x 1us = 10 ms

G10ms: LDX #1249 ;döngü sayacının ilk değeri

L2: DEX ;döngü sayacının


BNE L2 ;sıfır olana kadar azaltılması.

RTS ;ana programa geri dönülmesi.



Şekil 18-12 Mikroişlemci yazılımı kullanılarak zamanlama işaretlerinin üretilmesi

**Örnek 18-10** Aşağıda verilen iki darbe diyagramını, paralel çıkış birimi kullanarak oluşturan bir zamanlama yazılımı 32,768kHz osilatör hızında çalışan 6802 mikroişlemcisi dilinde tasarlanacaktır. Programı bir çevirici kaynak dosyası biçiminde olacak şekilde ve her satırındaki komutun açıklamasını yaparak yazınız

**Çözüm:**

```

PIADRB: EQU 8002H ;PIA Port B Veri/Veri Yönü (DRB/DDR) Yazmaç adresi
PIACRB: EQU 8003H ;PIA Port B Kontrol Yazmaç (CRB) adresi
ORG 0F000H
BASLA: LDS #7FH ;Yığın bölgesini iç RAM'in tepe adresine ayarla
CLR PIACRB ;PIA Port B Veri Yönü Yazmacının seçilmesi (CRB b2="0")
LDAA #0FFH
STAA PIADRB ;PIA Port B 8-Bit Çıkış (PB0-PB7=Çıkış) DDRB=00H
LDAA #04H
STAA PIACRB ;PIA Port B Veri Yazmacının seçilmesi (CRB b2="1")
ANA: LDAB #80H

```

```

[5] STAB PIADRB ; Q7="1"
[3] LDX #5117 ; 5 sn gecikme için 5117
[9] JSR GX ; gecikme altprogramı
[2] LDAB #00H

```

```

STAB PIADRB ; Q0-Q7="0"
LDX #1021 ; 1 sn gecikme için 1021
JSR GX ; gecikme altprogramı
LDAB #40H
STAB PIADRB ; Q6="1"
LDX #2045 ; 2 sn gecikme için 2045
JSR GX ; gecikme altprogramı
LDAB #00H
STAB PIADRB ; Q0-Q7="0"
LDX #1021 ; 1 sn gecikme için 1021
JSR GX ; gecikme altprogramı
JMP ANA ;darbe zamanlamasını tekrarlamak için başa dön

```

;1,2,5 sn gecikme sağlayan altprogram

;fXTAL= 32,768kHz ise sistem saati=8,192 kHz ise 1sn=8192 sistem saati

;zaman gecikmesi = 5 + 3 + 9 + TE x (4+4) + 5 + 2

;döngü sayacının ilk değeri 1sn TE=1021, 2sn TE=2045, 5sn TE=5117

```


GX: [4] DEX ;döngü sayacının
[4] BNE GX ;sıfır olana kadar azaltılması.
[5] RTS ;ana programa geri dönülmesi.
END

```






Eğer hiçbir tuşa basılmamışsa, sütunlar R direnci ile VCC'ye bağlı olduğu için PA4-PA6 uçlarından lojik "1" okunur.



Şekil 18-14 Paralel Giriş/Çıkış tümleşik devresine matris tipi tuş takımı bağlanması.



Şekil 18-15 Matris tipi tuş takımında basılan tuşun algılanması.

**Örnek 18-11** Şekil 18-14'de verilen şekilde 6821 PIA Paralel Giriş/Çıkış biriminin A portuna bağlı matris tipi bir tuş takımının hangi tuşuna basıldığını belirleyen 6802 mikroişlemci dilinde bir programı, çevirici çıkış dosyası şeklinde yazınız.

**Çözüm:**

8000 =	PIADRA: EQU	8000H ;PIA Port A Veri/Veri Yönü (DRA/DDRA) Yazmaç adresi
8001 =	PIACRA: EQU	8001H ;PIA Port A Kontrol Yazmaç (CRA) adresi
F000	ORG	0F000H
F000 8E007F	BASLA: LDS	#7FH ;Yığın bölgesini iç RAM'in tepe adresine ayarla
F003 7F8001	CLR	PIACRA ;PIA Port A Veri Yönü Yazmacının seçilmesi (CRA b2="0")
F006 860F	LDA	#0FH
F008 B78000	STAA	PIADRA ;PIA PA0-PA3=4-Bit Çıkış, PA4-PA7= Giriş, DDRA=0FH
F00B 8604	LDA	#04H
F00D B78001	STAA	PIACRA ;PIA Port A Veri Yazmacının seçilmesi (CRA b2="1")
F010 BDF100	ANA: JSR	BEK1 ;çinlamanın geçmesi için bekleme alt programı
F013 7F8000	CLR	PIADRA ;Sayısal çıkış birimine "0000" yazılması
F016 B68000	LDA	PIADRA ;Sayısal giriş biriminin okunması
F019 8470	ANDA	#01110000B;PA4-PA6 uçlarından okunan verinin elde edilmesi
F01B 8170	CMPA	#01110000B;Herhangi bir tuşa basıldı mı?
F01D 27F1	BEQ	ANA ;Tuşa basılmadıysa ana program döngüsüne git
F01F 860E	SAT1: LDA	#00001110B;Basılan tuş 1.satırda mı?
F021 B78000	STAA	PIADRA ;Sayısal çıkış birimine "1110" yazılması
F024 B68000	LDA	PIADRA ;Sayısal giriş biriminin okunması
F027 8470	ANDA	#01110000B;PA4-PA6 uçlarından okunan verinin elde edilmesi
F029 8160	CMPA	#01100000B;Basılan tuş "1" mi?
F02B 2606	BNE	S1K2 ;Basılan tuş "1" değilse S1K2'ye git
F02D BDE100	JSR	TUS1 ;Basılan tuş "1" ise yapılacak işlemler.
F030 7EF010	JMP	ANA ;ana program döngüsüne git
F033 8150	S1K2: CMPA	#01010000B;Basılan tuş "2" mi?
F035 2606	BNE	S1K3 ;Basılan tuş "2" değilse S1K3'e git
F037 BDE200	JSR	TUS2 ;Basılan tuş "2" ise yapılacak işlemler.
F03A 7EF010	JMP	ANA ;ana program döngüsüne git
F03D 8130	S1K3: CMPA	#00110000B;Basılan tuş "3" mü?
F03F 2606	BNE	SAT2 ;Basılan tuş "3" değilse SAT2'ye git
F041 BDE300	JSR	TUS3 ;Basılan tuş "3" ise yapılacak işlemler.
F044 7EF010	JMP	ANA ;ana program döngüsüne git
F047 860D	SAT2: LDA	#00001101B;Basılan tuş 2.satırda mı?
F049 B78000	STAA	PIADRA ;Sayısal çıkış birimine "1101" yazılması
F04C B68000	LDA	PIADRA ;Sayısal giriş biriminin okunması
F04F 8470	ANDA	#01110000B;PA4-PA6 uçlarından okunan verinin elde edilmesi
F051 8160	CMPA	#01100000B; Basılan tuş "4" mü?
F053 2606	BNE	S2K2 ;Basılan tuş "4" değilse S2K2'ye git
F055 BDE400	JSR	TUS4 ;Basılan tuş "4" ise yapılacak işlemler.
F058 7EF010	JMP	ANA ;ana program döngüsüne git
F05B 8150	S2K2: CMPA	#01010000B;Basılan tuş "5" mi?
F05D 2606	BNE	S2K3 ;Basılan tuş "5" değilse S2K3'e git
F05F BDE500	JSR	TUS5 ;Basılan tuş "5" ise yapılacak işlemler.
F062 7EF010	JMP	ANA ;ana program döngüsüne git
F065 8130	S2K3: CMPA	#00110000B;Basılan tuş "6" mi?
F067 2606	BNE	SAT3 ;Basılan tuş "6" değilse SAT3'e git
F069 BDE600	JSR	TUS6 ;Basılan tuş "6" ise yapılacak işlemler.
F06C 7EF010	JMP	ANA ;ana program döngüsüne git

F06F 860B	SAT3:	LDAA	#00001011B;	Basılan tuş 3.satırda mı?
F071 B78000		STAA	PIADRA	;Sayısal çıkış birimine "1011" yazılması
F074 B68000		LDAA	PIADRA	;Sayısal giriş biriminin okunması
F077 8470		ANDA	#01110000B;	PA4-PA6 uçlarından okunan verinin elde edilmesi
F079 8160		CMPA	#01100000B;	Basılan tuş "7" mi?
F07B 2606		BNE	S3K2	;Basılan tuş "7" değilse S3K2'ye git
F07D BDE700		JSR	TUS7	;Basılan tuş "7" ise yapılacak işlemler.
F080 7EF010		JMP	ANA	;ana program döngüsüne git
F083 8150	S3K2:	CMPA	#01010000B;	Basılan tuş "8" mi?
F085 2606		BNE	S3K3	;Basılan tuş "8" değilse S3K3'e git
F087 BDE800		JSR	TUS8	;Basılan tuş "8" ise yapılacak işlemler.
F08A 7EF010		JMP	ANA	;ana program döngüsüne git
F08D 8130	S3K3:	CMPA	#00110000B;	Basılan tuş "9" mu?
F08F 2606		BNE	SAT4	;Basılan tuş "9" değilse SAT4'e git
F091 BDE900		JSR	TUS9	;Basılan tuş "9" ise yapılacak işlemler.
F094 7EF010		JMP	ANA	;ana program döngüsüne git

---

F097 8607	SAT4:	LDAA	#00000111B;	Basılan tuş 4.satırda mı?
F099 B78000		STAA	PIADRA	;Sayısal çıkış birimine "0111" yazılması
F09C B68000		LDAA	PIADRA	;Sayısal giriş biriminin okunması
F09F 8470		ANDA	#01110000B;	PA4-PA6 uçlarından okunan verinin elde edilmesi
F0A1 8160		CMPA	#01100000B;	Basılan tuş "*" mı?
F0A3 2606		BNE	S4K2	;Basılan tuş "*" değilse S4K2'ye git
F0A5 BDEA00		JSR	TUSY	;Basılan tuş "*" ise yapılacak işlemler.
F0A8 7EF010		JMP	ANA	;ana program döngüsüne git
F0AB 8150	S4K2:	CMPA	#01010000B;	Basılan tuş "0" mı?
F0AD 2606		BNE	S4K3	;Basılan tuş "0" değilse S4K3'e git
F0AF BDE000		JSR	TUS0	;Basılan tuş "0" ise yapılacak işlemler.
F0B2 7EF010		JMP	ANA	;ana program döngüsüne git
F0B5 8130	S4K3:	CMPA	#00110000B;	Basılan tuş "#" mi?
F0B7 2603		BNE	SATX	;Basılan tuş "#" değilse SATX'e git
F0B9 BDEB00		JSR	TUSD	;Basılan tuş "#" ise yapılacak işlemler.
F0BC 7EF010	SATX:	JMP	ANA	;ana program döngüsüne git

;10 ms gecikme sağlayan alt program, sistem saati 1 MHz ise  $10000 \times 1\mu s = 10 \text{ ms}$ .

;6802 için toplam gecikme =  $3 + 1249 \times (4+4) + 5 = 10000$  sistem saati


F100		ORG	0F100H	
F100 CE04E1	BEK1:	LDX	#1249	;döngü sayacının ilk değeri
F103 09	L2:	DEX		;döngü sayacının
F104 26FD		BNE	L2	;sıfır olana kadar azaltılması.
F106 39		RTS		;ana programa geri dönülmesi.

;Altprogramlar


E000		ORG	0E000H	
E000 39	TUS0:	RTS		
E100		ORG	0E100H	
E100 39	TUS1:	RTS		
E200		ORG	0E200H	
E200 39	TUS2:	RTS		
E300		ORG	0E300H	
E300 39	TUS3:	RTS		
E400		ORG	0E400H	
E400 39	TUS4:	RTS		

```
E500          ORG  0E500H
E500 39      TUS5: RTS
E600          ORG  0E600H
E600 39      TUS6: RTS
E700          ORG  0E700H
E700 39      TUS7: RTS
E800          ORG  0E800H
E800 39      TUS8: RTS
E900          ORG  0E900H
E900 39      TUS9: RTS
EA00          ORG  0EA00H
EA00 39      TUSY: RTS
EB00          ORG  0EB00H
EB00 39      TUSD: RTS
              ;Vektör Adresleri
FFFE          ORG  0FFFEH
FFFE F000    DWM  BASLA ;RES Reset, Mikroişlemciyi Yeniden Başlatma Adresi
0000          END
```


## 18.5. Çok Basamaklı Gösterge Uygulamaları




Şekil 18-16 Çok sayıda ortak anot göstergenin bir PIA ile sürülmesi



Şekil 18-17 Çok basamaklı 7-parça göstergelerin dinamik tarama ile sürülmesi



Şekil 18-18 Bir 4 basamaklı 7-parça göstergenin dinamik tarama ile sürülmesi



Şekil 18-19 Sürücü Blokların Açık Devresi ve Gösterge Bağlantısı

**Örnek 18-12** Şekil 18-16'daki devre için verilen şekilde göstergede sırayla 1'den 4'e kadar olan sayıların "1234" şeklinde görünmesini sağlayan 6802 mikroişlemci dilinde bir programı çevirici çıkış dosyası şeklinde yazınız ve çalışmasını kısaca açıklayınız.

**Çözüm:****Programın çıkış dosyası:**

```

8000 =      PIADRA: EQU   8000H ;PIA Port A Veri/Veri Yönü (DRA/DDRA) Yazmaç adresi
8001 =      PIACRA: EQU   8001H ;PIA Port A Kontrol Yazmaç (CRA) adresi
8002 =      PIADRB: EQU  8002H ;PIA Port B Veri/Veri Yönü (DRB/DDRB) Yazmaç adresi
8003 =      PIACRB: EQU  8003H ;PIA Port B Kontrol Yazmaç (CRB) adresi
F000      ORG    0F000H
F000 8E007F BASLA: LDS   #7FH   ;Yığın bölgesini iç RAM'in tepe adresine ayarla
F003 7F8001      CLR   PIACRA ;PIA Port A Veri Yönü Yazmacının seçilmesi (CRA b2="0")
F006 7F8003      CLR   PIACRB ;PIA Port B Veri Yönü Yazmacının seçilmesi (CRB b2="0")
F009 86FF      LDAA  #0FFH
F00B B78000      STAA  PIADRA ;PIA Port A=8-Bit Çıkış, PA0-PA7=Çıkış, DDRA=FFH
F00E B78002      STAA  PIADRB ;PIA Port B=8-Bit Çıkış, PB0-PB7=Çıkış DDRB=FFH
F011 8604      LDAA  #04H
F013 B78001      STAA  PIACRA ;PIA Port A Veri Yazmacının seçilmesi (CRA b2="1")
F016 B78003      STAA  PIACRB ;PIA Port B Veri Yazmacının seçilmesi (CRB b2="1")
-----
F019 8601      ANA:   LDAA  #0000001B ;göstergenin 1.basamağının seçilmesi
F01B B78002      STAA  PIADRB ;Sayısal çıkış biriminin B portuna yazılması
F01E 8666      LDAA  #66H   ;göstergeye "4" bilgisi için kod
F020 B78000      STAA  PIADRA ;Sayısal çıkış biriminin A portuna yazılması
F023 BDF100      JSR   BEK1   ;yazılanı gözün algılaması için bekleme altprogramı
-----
F026 8602      LDAA  #00000010B ;göstergenin 2.basamağının seçilmesi
F028 B78002      STAA  PIADRB ;Sayısal çıkış biriminin B portuna yazılması
F02B 864F      LDAA  #4FH   ;göstergeye "3" bilgisi için kod
F02D B78000      STAA  PIADRA ;Sayısal çıkış biriminin A portuna yazılması
F030 BDF100      JSR   BEK1   ;yazılanı gözün algılaması için bekleme altprogramı
-----
F033 8604      LDAA  #00000100B ;göstergenin 3.basamağının seçilmesi
F035 B78002      STAA  PIADRB ;Sayısal çıkış biriminin B portuna yazılması
F038 865B      LDAA  #5BH   ;göstergeye "2" bilgisi için kod
F03A B78000      STAA  PIADRA ;Sayısal çıkış biriminin A portuna yazılması
F03D BDF100      JSR   BEK1   ;yazılanı gözün algılaması için bekleme altprogramı
-----
F040 8608      LDAA  #00001000B ;göstergenin 4.basamağının seçilmesi
F042 B78002      STAA  PIADRB ;Sayısal çıkış biriminin B portuna yazılması
F045 8606      LDAA  #06H   ;göstergeye "1" bilgisi için kod
F047 B78000      STAA  PIADRA ;Sayısal çıkış biriminin A portuna yazılması
F04A BDF100      JSR   BEK1   ;yazılanı gözün algılaması için bekleme altprogramı
-----
F04D 7EF019      JMP   ANA    ;ana program döngüsüne git
;10 ms gecikme sağlayan alt program, sistem saati 1 MHz ise 10000 x 1us = 10 ms.
;6802 için toplam gecikme = 3 + 1249 x (4+4) + 5 = 10000 sistem saati
F100      ORG    0F100H
F100 CE04E1      BEK1: LDX   #1249 ;döngü sayacının ilk değeri
F103 09      L2:   DEX           ;döngü sayacının
F104 26FD      BNE   L2      ;sıfır olana kadar azaltılması.
F106 39      RTS           ;ana programa geri dönülmesi.
;Vektör Adresleri
FFFE      ORG    0FFFEH
FFFE F000      DWM   BASLA ;RES Reset, Mikroişlemciyi Yeniden Başlatma Adresi
0000      END

```

**Örnek 18-13** Şekil 18-16'daki devre için verilen şekilde göstergede sırayla 1'den 4'e kadar olan sayıların "1234" şeklinde görünmesini sağlayan 6802 mikroişlemci dilinde bir program yazınız.

**Çözüm:** Burada çözüm olarak Örnek 18-12'deki çözümdekine benzer şekilde bir donanım ve yazılım kullanılabilir. Bu çözümdeki programın çizgilerle belirlenmiş olan her basamak için tekrarlanan kısmı tabloya bakma yöntemiyle tasarlanarak istenen sonuç elde edilir.

**Programın çıkış dosyası:**

```


8000 =      PIADRA: EQU  8000H ;PIA Port A Veri/Veri Yönü (DRA/DDRA) Yazmaç adresi
8001 =      PIACRA: EQU  8001H ;PIA Port A Kontrol Yazmaç (CRA) adresi
8002 =      PIADRB: EQU  8002H ;PIA Port B Veri/Veri Yönü (DRB/DDRB) Yazmaç adresi
8003 =      PIACRB: EQU  8003H ;PIA Port B Kontrol Yazmaç (CRB) adresi
0000      ORG  0000H
0000      TMP:  DFS  2      ;Geçici veri için bellekte 2 Bayt yer ayr
F000      ORG  0F000H
F000 8E007F BASLA: LDS  #7FH ;Yığın bölgesini iç RAM'in tepe adresine ayarla
F003 7F8001 CLR  PIACRA ;PIA Port A Veri Yönü Yazmacının seçilmesi (CRA b2="0")
F006 7F8003 CLR  PIACRB ;PIA Port B Veri Yönü Yazmacının seçilmesi (CRB b2="0")
F009 86FF   LDAA #0FFH
F00B B78000 STAA PIADRA ;PIA Port A=8-Bit Çıkış, PA0-PA7=Çıkış, DDRA=FFH
F00E B78002 STAA PIADRB ;PIA Port B=8-Bit Çıkış, PB0-PB7=Çıkış DDRB=FFH
F011 8604   LDAA #04H
F013 B78001 STAA PIACRA ;PIA Port A Veri Yazmacının seçilmesi (CRA b2="1")
F016 B78003 STAA PIACRB ;PIA Port B Veri Yazmacının seçilmesi (CRB b2="1")
F019 C608   ANA:  LDAB #00001000B ;göstergenin 4.basamağının seçilmesi
F01B CEF200 LDX  #GTBL ;gösterge değerleri için tablonun başlangıç adresi
F01E A600   L1:  LDAA 0,X ;gösterge bilgisi için kodun tablodan alınması
F020 B78000 STAA PIADRA ;Sayısal çıkış biriminin A portuna yazılması
F023 F78002 STAB PIADRB ;basamak seçimi için PIA'nın B portuna yazılması
F026 DF00   STX  TMP ;X dizin yazmacının bellekte saklanması
F028 BDF100 JSR  BEK1 ;yazılanı gözün algılaması için bekleme alt programı
F02B DE00   LDX  TMP ;X dizin yazmacının bellekten geri alınması
F02D 08     INX ;bir sonraki gösterge bilgisi adresi
F02E 54     LSRB ;bir sonraki basamağın seçimi
F02F 26ED   BNE  L1 ;sayıların gösterilmesi bitene kadar devam.
F031 7EF019 JMP  ANA ;ana program döngüsüne geri dön
F100      ORG  0F100H
;10 ms gecikme sağlayan alt program, sistem saati 1 MHz ise 10000 x 1us = 10 ms.
;6802 için toplam gecikme = 3 + 1249 x (4+4) + 5 = 10000 sistem saati
F100 CE04E1 BEK1: LDX  #1249 ;döngü sayacının ilk değeri
F103 09     L2:  DEX ;döngü sayacının
F104 26FD   BNE  L2 ;sıfır olana kadar azaltılması.
F106 39     RTS ;ana programa geri dönülmesi.
F200      ORG  0F200H
;Gösterge kodlarının bulunduğu tablo
F200 065B4F66GTBL: DFB  06H,5BH,4FH,66H ;"1","2","3","4"
;Vektör Adresleri
FFFE      ORG  0FFFEH
FFFE F000   DWM  BASLA ;RES Reset, Mikroişlemciyi Yeniden Başlatma Adresi
0000      END

```



**Örnek 18-14** Dinamik gösterge tarama işlemiyle çalışan bir program ile 4 basamaklı ortak katotlu 7-parça LED göstergeye sayısal çıkış işlemi gerçekleştirilecektir. Bunun için bir çıkış birimi sistemini tasarlayınız ve blok diyagramını çiziniz.


**Çözüm:**



Şekil 18-20 Çok sayıda ortak katot göstergenin bir PIA ile sürülmesi

## 18.6. Paralel Yazıcı Uygulaması

Gündelik yaşamdaki yazarkasa fişi, fatura gibi birçok uygulamada belgelere çıkış için yazıcılar kullanılmaktadır. Eğer yazıcı mikroişlemcili sistemden veriyi paralel olarak transfer ediyorsa bu çeşit yazıcılar paralel yazıcı olarak adlandırılır.



Şekil 18-21 Paralel Giriş/Çıkış tümleşik devresinin yazıcı arabirim için kullanılması

**Örnek 18-15** Şekil 18-21'deki blok diyagramda gösterilen şekilde 6821 PIA tümleşik devresiyle centronics arabirimli bir paralel yazıcı arabirimi tasarlanacaktır. Yazıcı genel durum bilgileri için PIA Port A 4-Bit giriş, yazıcı veri bilgisi için Port B 8-Bit çıkış ve Port B kontrol uçları belirteç (STB), alındı (ACK) olarak kullanılacaktır.

- Kontrol yazmaçlara yazılacak kontrol kelimesini kısaca açıklayınız.
- ROM'da saklanan bir mesajı yazıcıya yazdıran bir programı 6802 mikroişlemcisi dilinde, bir çevirici kaynak dosyası biçiminde olacak şekilde gerekli olan bütün tanımlamaları ve her satırındaki komutun açıklamasını yaparak yazınız.

### Çözüm:

Önce Port A okunacak ve "ERROR, PE, SLCT, BUSY" gibi yazıcı durum işaretlerine bakılacaktır. Paralel yazıcıya yazılmasıyla CB2 kontrol ucundan "Belirteç" STB adıyla bir "0" çıkış işareti üretilecek ve yazıcıdan CB1 kontrol ucundan "Alındı" ACK adıyla düşen kenar alındıktan sonra CB2 yeniden "1" olacaktır. ACK işareti, mikroişlemciye kesme işareti üretecek ve kesme hizmet programıyla yazıcıya veri yazılacaktır.


#### a) Kontrol kelimesinin açıklaması:

Yazıcıya yazılmasıyla CB2 kontrol ucundan "STB" adıyla bir çıkış işareti üretilmesi CB1 ucuyla "ACK" ile yenilenen Yazma belirteci kullanılarak sağlanabilir.

	b7	b6	b5	b4	b3	b2	b1	b0
<b>CRB=25H</b>	0	0	1	0	0	1	0	1
	IRQB1	IRQB2	CB2 Kontrol		DDR	CB1 Kontrol		

CB1 kontrol ucu düşen kenar giriş ve mikroişlemciye kesme işaretine izin  
 CB2 kontrol ucu çıkış ve CB1 ucuyla yenilenen Yazma belirteci  
 b2=1 Port B veri yazmaya hazır

#### Çalışma zamanlama diyagramı:



#### b) Programın kaynak dosyası:

;PIA iç yazmaç adres tanımlamaları

```
PIADRA: EQU 8000H ;PIA Port A Veri/Veri Yönü (DRA/DDRA) Yazmaç adresi
PIACRA: EQU 8001H ;PIA Port A Kontrol Yazmaç (CRA) adresi
PIADRB: EQU 8002H ;PIA Port B Veri/Veri Yönü (DRB/DDRB) Yazmaç adresi
PIACRB: EQU 8003H ;PIA Port B Kontrol Yazmaç (CRB) adresi
```

ORG 0000H ;RAM bellek değişken tanımları

TMP: DFS 1 ;Geçici veri için bellekte 1 Bayt yer ayır

ORG 0F000H ;İlk koşullama

BASLA: LDS #7FH ;Yığın bölgesini iç RAM'in tepe adresine ayarla

CLR PIACRA ;PIA Port A Veri Yönü Yazmacının seçilmesi (CRA b2="0")

CLR PIACRB ;PIA Port B Veri Yönü Yazmacının seçilmesi (CRB b2="0")

CLRA

STAA PIADRA ;PIA Port A=8-Bit Giriş, PA0-PA7=Giriş, DDRA=00H

LDAA #0FFH

STAA PIADRB ;PIA Port B=8-Bit Çıkış, PB0-PB7=Çıkış DDRB=FFH

LDAA #25H

STAA PIACRA ;PIA Port A Veri Yazmacının seçilmesi (CRA b2="1")

STAA PIACRB ;PIA Port B Veri Yazmacının seçilmesi (CRB b2="1")

;Ana Program

ANA: JSR YAZDIR ;Mesaj yazdır altprogramını çağır


JMP ANA ;ana program döngüsüne geri dön

```

ORG 0F100H ;Yazdır altprogramı
YAZDIR: CLI ;Mikroişlemcide IRQ kesmesine izin
LDX #MES1 ;Yazılacak yazının başlangıç adresi
L1: JSR HATA ;Yazıcı durumuna bak
BCS L1 ;Yazıcı hazır değilse bekle
DYAZ: WAI ;Donanım kesmesi (IRQ) bekle
LDAA TMP ;yazılacak veriyi geçici bellekten A akümülatörüne yükle
CMPA #0AH ;CR+LF, Son veri yazıldı mı?
BEQ SON ;yazdırma işlemi bittiğinde altprogramın sonuna git
INX
BRA DYAZ ;yazdırma işlemi bitene kadar devam
SON: RTS ;YAZDIR altprogramının sonu
ORG 0F200H ;Yazıcı kesme hizmet programı
KESME: LDAA 0,X ;X+0 adresinden yazılacak veriyi oku
STAA PIADRB ;veriyi yazıcıya gönder
STAA TMP ;veriyi sakla
RTI ;kesme hizmet programından geri dönüş
ORG 0F200H ;Yazılacak Mesajın ASCII kodlarının bulunduğu tablo
MES1: DFB "Bu bir deneme yazısıdır.",0DH,0AH
ORG 0F300H ;Yazıcı hataları altprogramı
HATA: LDAA PIADRA ;Yazıcı durumuna bak
LSRA
BCC HVAR ;ERROR=0, Yazıcı hazır değil, hata var!
LSRA
BCC HVAR ;SLCT=0, Yazıcı Hazır değil!
LSRA
BCS HVAR ;PE=1, Yazıcıda kağıt yok!
LSRA
BCS HVAR ;BUSY=1, Yazıcı meşgul!
RTS ;hata yok! C=0
HVAR: SEC ;Hata var! C=1
RTS ;HATA altprogramının sonu
ORG 0FFF8H ;Vektör Adresleri
DWM KESME ;IRQ Örtülebilir Kesme, Yazıcı kesme hizmet programı
DWM BASLA ;SWI Yazılım Kesmesi
DWM BASLA ;NMI Örtülemez Kesme
DWM BASLA ;RES Reset, Mikroişlemciyi Yeniden Başlatma Adresi
END


```

## 18.7. Analog Sayısal Dönüştürücü Uygulamaları




Şekil 18-22 Analog giriş uygulaması

**Örnek 18-16** Şekil 18-22'de gösterilen şekilde 6821 PIA tümleşik devresiyle Port A kullanılarak aşağıda "Başla" START ve "Dönüşüm Bitti" EOC çalışma zamanlama diyagramı verilen 8-Bit ADC birimi 6802 mikroişlemcili sisteme bağlanacaktır.



## 18.8. Sayısal Analog Dönüştürücü Uygulaması



Şekil 18-23 Gerilim çıkışlı ikili girişli sayısal analog dönüştürücü birimi


$V_{OUT}$  çıkış geriliminin sayısal giriş ve REF+ ucuna uygulanan gerileme bağılı ifadesi aşağıda verilen şekildedir.

$$V_{OUT} = 10V \left( \frac{D7}{2} + \frac{D6}{4} + \frac{D5}{8} + \frac{D4}{16} + \frac{D3}{32} + \frac{D2}{64} + \frac{D1}{128} + \frac{D0}{256} \right)$$

En küçük analog çıkış gerilimi için  $D0-D7="0"$  dir ve çıkış=0V olur.

En büyük analog çıkış gerilimi için  $D0-D7="1"$  dir ve çıkış $\cong$ 10V olur.


$$V_{OUT} = 10V \left( \frac{1}{2} + \frac{1}{4} + \frac{1}{8} + \frac{1}{16} + \frac{1}{32} + \frac{1}{64} + \frac{1}{128} + \frac{1}{256} \right) = 10 \frac{255}{256} = 9,96V$$



Şekil 18-24 Gerilim çıkışlı ikili girişli sayısal analog dönüştürücü birimi


**Örnek 18-17** Şekil 18-24'de gösterildiği gibi 6821 PIA tümleşik devresiyle 8-Bit DAC birimi kullanarak 100Hz frekanslı bir testere dişi işaret üretilecektir. Programı 6802 mikroişlemcisinin dilinde, bir çevirici kaynak dosyası biçiminde olacak şekilde gerekli olan bütün tanımlamaları ve her satırındaki komutun açıklamasını yaparak yazınız.

Mikroişlemci temelli sistemlerde yaygın olarak kullanılan diğer bir sayısal işareti analog işarete dönüştürme yöntemi ise darbe genişlik modülasyonu (PWM) yöntemidir.



Şekil 18-25 Darbe genişliği modülasyonu

## 18.9. Birden Fazla Kesme İşaretinin Mikroişlemciye Uygulanması



Şekil 18-26 Birden fazla kesme işaretinin mikroişlemciye bağlanması

```


BASLA: .... ;başlangıçta yapılacak işlemler.
        .... ;
        CLI ;örtülebilir kesme isteğine izin
;Ana Program
ANA: .... ;ana programda yapılacak işlemler.
     .... ;
     JMP ANA ;ana program döngüsüne git
;IRQ örtülebilir kesme isteği için servis programı
KESME: LDA 0D900H ;kesme kaynağı için giriş biriminin okunması
        AND #00000111B ;D0-D3 uçlarından okunan verinin elde edilmesi
IRQ1:
        CMA #0000001B ;kaynak IRQ1 mi?
        BNE IRQ2
        .... ;IRQ1 ise yapılacak işlemler.
        RTI ;ana program döngüsüne geri dön
IRQ2:
        CMA #0000010B ;kaynak IRQ2 mi?
        BNE IRQ3
        .... ;IRQ2 ise yapılacak işlemler.
        RTI ;ana program döngüsüne geri dön
IRQ3:
        CMA #00000100B ;kaynak IRQ3 mü?
        BNE IRQX
        .... ;IRQ3 ise yapılacak işlemler.
IRQX: RTI ;ana program döngüsüne geri dön
      ORG 0FFF8H ;Vektör Adresleri
      DWM KESME ;IRQ Örtülebilir Kesme Servis Program Adresi
      DWM BASLA ;SWI Yazılım İle Kesme Servis Program Adresi
      DWM BASLA ;NMI Örtülemez Kesme Servis Program Adresi
      DWM BASLA ;RES Reset, Mikroişlemciyi Yeniden Başlatma Adresi
      END

```


## 18.10. Genel Amaçlı Bir 6802 Mikroşlemcili Sistem Uygulaması

Günümüzde eğitim, endüstriyel, askeri, tıbbi, vs. birçok uygulamayı kolayca gerçekleştirebilecek çevre birimleri donanımına sahip genel amaçlı mikroşlemci temelli sistemler ticari amaçlı olarak mikroşlemciyi üreten firma veya bu uygulama alanlarında üretim yapan uzmanlaşmış firmalar tarafından tasarlanarak üretilmektedir. Şekil 18-27'de mikroşlemci ve adres çözümleme birimi, Şekil 18-28'de ana bellek birimi, Şekil 18-29'de bellek haritası, Şekil 18-30'da paralel G/Ç, tuş takımı ve gösterge birimi, Şekil 18-31'de ise seri G/Ç, baud oranı üretici ve RS-232 arabirim donanımları verilmiştir.


Şekil 18-27 6802 Mikroişlemci ve Adres Çözümleme Birimi








Şekil 18-28 Program ve Veri Belleği Birimi



Şekil 18-29 Genel Amaçlı Sistemin Bellek Haritası





Şekil 18-31 Seri G/Ç Birimi, Baud Oranı Üretici ve RS-232 Sürücü Devreleri